2. Vorlesung Mikroskopische Bildverarbeitung

Wahlpflichtmodul 9521: EI-M im 1. und 3. Fachsemester

Initialisierung

6. November 2015

Wiederholung von der 1. Vorlesung

Aus 5. Bilderfassung (Abtastung)

Bildquantisierung (“Treppenfunktion”)

```
Manipulate[
 Show[Image[Round[Round[ImageData[phasenkontrast, "Byte"], 255 / (2^bit - 1)]], "Byte"], ImageSize → ImageDimensions[phasenkontrast]],
 {{bit, 3, "Bit"}, 1, 8, 1, Appearance → "Labeled"}, ContinuousAction → False, SaveDefinitions → True]
```

Bit 3



Bilddiskretisierung (örtlich)

```
(*behält Originalrastergröße*)
Manipulate[
 Show[Image[ArrayFlatten[Map[Function[Table[#, {2^oktaven}, {2^oktaven}]][Mean[Flatten[#]]] &, Partition[ImageData[phasenkontrast],
       {2^oktaven, 2^oktaven}, {2^oktaven, 2^oktaven}], {2}]]], ImageSize → ImageDimensions[phasenkontrast]],
 {{oktaven, 2, "Oktaven"}, 0, Floor@Log[2, Min@ImageDimensions[phasenkontrast]], 1, Appearance → "Labeled"},
 ContinuousAction → False, SaveDefinitions → True]
```

Oktaven 4



```
(*paßt Rastergröße an*)
Manipulate[Show[Image[Partition[Map[Plus @@ Flatten[#] * N[1 / (Times @@ {2^oktaven, 2^oktaven})] &,
      Flatten[Partition[ImageData[phasenkontrast], {2^oktaven, 2^oktaven}, {2^oktaven, 2^oktaven}], 1]],
    Floor[First[ImageDimensions[phasenkontrast]] / (2^oktaven)]]], ImageSize → ImageDimensions[phasenkontrast]],
 {{oktaven, 2, "Oktaven"}, 0, Floor@Log[2, Min@ImageDimensions[phasenkontrast]], 1, Appearance → "Labeled"},
 ContinuousAction → False, SaveDefinitions → True]
```

Oktaven 4



Aus 6. Farbe

RGB

```
Join[{mauskniergb}, ColorSeparate[mauskniergb, "RGB"]]
```



```
ListPointPlot3D[Flatten[ImageData[mauskniergb], 1][[ ;; ;; 100]], ColorFunction → Function[{r, g, b}, RGBColor[r, g, b]],
 AxesLabel → {Text[Style["R", Red, 12]], Text[Style["G", Green, 12]], Text[Style["B", Blue, 12]]},
 ColorFunctionScaling → False, ImageSize → Medium, BoxRatios → {1, 1, 1}, PlotRange → {{0, 1}, {0, 1}, {0, 1}}]
```



```
Join[{ColorCombine[{mauskniey, mauskniey, mauskniey}, "RGB"]}, ColorSeparate[ColorCombine[{mauskniey, mauskniey, mauskniey}, "RGB"], "RGB"]]
```



```
ListPointPlot3D[Flatten[ImageData[ColorCombine[{mauskniey, mauskniey, mauskniey}, "RGB"]], 1][[ ;; ;; 100]],
 ColorFunction → Function[{r, g, b}, RGBColor[r, g, b]],
 AxesLabel → {Text[Style["R", Red, 12]], Text[Style["G", Green, 12]], Text[Style["B", Blue, 12]]},
 ColorFunctionScaling → False, ImageSize → Medium, BoxRatios → {1, 1, 1}, PlotRange → {{0, 1}, {0, 1}, {0, 1}}]
```



YC_bC_r

```
cm = {{0.299, 0.587, 0.114}, {-0.16874, -0.33126, 0.5}, {0.5, -0.41869, -0.08131}};
icm = Inverse@cm;
Join[{mauskniergb},
 Image[Partition[#, First[ImageDimensions[mauskniergb]]]] & /@ (cm.Transpose@Flatten[ImageData[mauskniergb], 1] + {0, 0.5, 0.5})]
```



```
ListPointPlot3D[Transpose[cm.Transpose@Flatten[ImageData[mauskniergb], 1]][[ ;; ;; 100]],
 ColorFunction → Function[{y, cb, cr}, (RGBColor @@ (icm.{y, cb, cr}))], ColorFunctionScaling → False,
 AxesLabel → {Text[Style["Y", 12]], Text[Style["Cb", 12]], Text[Style["Cr", 12]]},
 ImageSize → Medium, BoxRatios → {1, 1, 1}, PlotRange → {{0, 1}, {-0.5, 0.5}, {-0.5, 0.5}}]
```



```
Join[{mauskniey}, Image[Partition[#, First[ImageDimensions[mauskniey]]]] & /@
  (cm.Transpose@Flatten[ImageData[ColorCombine[{mauskniey, mauskniey, mauskniey}, "RGB"]], 1] + {0, 0.5, 0.5})]
```



```
{mauskniergb, mausknieg = ColorConvert[mauskniergb, "Grayscale"]}
```



```
{mauskniergbdata, mauskniegdata} = Flatten[ImageData[#, "Real"], 1] & /@ {mauskniergb, mausknieg};

koeffizienten = LeastSquares[mauskniergbdata, mauskniegdata]
```

```
{0.298958, 0.585863, 0.114866}
```

```
ImageApply[{##}.koeffizienten &, mauskniergb]
```



CMYK

```
Clear[rgbtocmyk];
rgbtocmyk[{r_Real, g_Real, b_Real}] := {1. - r, 1. - g, 1. - b, 1. - Max[{r, g, b}]};

Clear[cmyktorgb];
cmyktorgb[{c_Real, m_Real, y_Real, k_Real}] := {(1. - c), (1. - m) , (1. - y) };

rgbtocmyk[{.2, .2, .2}]
```

```
{0.8, 0.8, 0.8, 0.8}
```

```
Join[{mauskniergb}, Image[#] & /@ Transpose[Map[rgbtocmyk[#] &, ImageData[mauskniergb], {2}], {2, 3, 1}]]
```



```
ListPointPlot3D[Map[rgbtocmyk[#] &, Flatten[ImageData[mauskniergb], 1]][[All, 1 ;; 3]][[ ;;  ;; 100]],
 ColorFunction → Function[{c, m, y}, (RGBColor @@ (cmyktorgb@{c, m, y, .5}))], ColorFunctionScaling → False,
 AxesLabel → {Text[Style["C", Cyan, 12]], Text[Style["M", Magenta, 12]], Text[Style["Y", Yellow, 12]]},
 ImageSize → Medium, BoxRatios → {1, 1, 1}, PlotRange → {{0, 1}, {0, 1}, {0, 1}}]
```



```
Join[{mauskniey},
 Image[#] & /@ Transpose[Map[rgbtocmyk[#] &, ImageData[ColorCombine[{mauskniey, mauskniey, mauskniey}, "RGB"]], {2}], {2, 3, 1}]]
```



Tonwertkurve linear-potentiell (Variante 1): $o=c\,(b+i^{\gamma})$

```
Manipulate[Grid[
  {{Plot[c * (b + i^γ), {i, 0, 1}, PlotRange → {{0, 1}, {0, 1}}, PlotStyle → {Black}, AspectRatio → 1, ImageSize → ImageDimensions[zellealginat]],
    Show[Image[c * (b + ImageData[zellealginat, "Real"]^γ)], ImageSize → ImageDimensions[zellealginat]]}}],
 {{c, 1, "c"}, -2., 2., .1, Appearance → "Labeled"}, {{b, 0, "b"}, -1., 1., .1, Appearance → "Labeled"},
 {{γ, .5, "γ"}, .1, 10, .1, Appearance → "Labeled"}, ContinuousAction → False, SaveDefinitions → True]
```



Tonwertkurve logarithmisch: $o=\frac{\log(i\, b^a+1)}{\log(b^a+1)}$

```
Manipulate[Grid[{{Plot[Log[1. + b^a * i] / Log[1. + b^a], {i, 0, 1},
      PlotRange → {{0, 1}, {0, 1}}, PlotStyle → {Black}, AspectRatio → 1, ImageSize → ImageDimensions[zellealginat]],
    Show[Image[Log[1. + b^a * ImageData[zellealginat, "Real"] / Log[1. + b^a]]], ImageSize → ImageDimensions[zellealginat]]}}],
 {{a, 1, "a"}, 1. / 16, 4., 1 / 16, Appearance → "Labeled"}, {{b, 2, "b"}, 2., 10., 1, Appearance → "Labeled"},
 SaveDefinitions → True]
```



Tonwertkurve logistisch: $o=\frac{1}{e^{-a\left(i-\frac{1}{2}\right)}+1}$ (im Funktions- und Wertebereich 0...1)

```
Manipulate[Grid[{{Plot[1 / (1 + Exp[-a * (i - .5)]), {i, 0, 1},
      PlotRange → {{0, 1}, {0, 1}}, PlotStyle → {Black}, AspectRatio → 1, ImageSize → ImageDimensions[zellealginat]],
    Show[Image[1 / (1 + Exp[-a * (ImageData[zellealginat, "Real"] - .5))]], ImageSize → ImageDimensions[zellealginat]]}}],
 {{a, 5., "Steilheit a"}, 2., 50., 1., Appearance → "Labeled"}, SaveDefinitions → True]
```

Steilheit a 5.





Tonwertkurve linear-potentiell (Variante 2): $o = (c+1)\left((b+1)\, i^{\gamma} - \frac{1}{2}\right) + \frac{1}{2}$

```
Manipulate[Grid[{{Plot[Clip[(c + 1) ((b + 1) i^γ - 1 / 2) + 1 / 2, {0, 1}], {i, 0, 1},
      PlotRange → {{0, 1}, {0, 1}}, PlotStyle → {Black}, AspectRatio → 1, ImageSize → ImageDimensions[zellealginat]],
    Show[Image[Clip[(c + 1) ((b + 1) ImageData[zellealginat, "Real"]^γ - 1 / 2) + 1 / 2, {0, 1}]], ImageSize → ImageDimensions[zellealginat]]}}],
 {{c, 2., "c"}, 0., 10., .1, Appearance → "Labeled"}, {{b, 0.1, "b"}, 0., 1., .1, Appearance → "Labeled"},
 {{γ, .8, "γ"}, .1, 2, .1, Appearance → "Labeled"}, SaveDefinitions → True]
```



Variante 2 ist in ImageAdjust von *Mathematica* eingebaut:

```
Manipulate[
 Grid[{{Plot[Clip[(c + 1) ((b + 1) i^γ - 1 / 2) + 1 / 2, {0, 1}], {i, 0, 1}, PlotRange → {{0, 1}, {0, 1}}, PlotStyle → {Black}, AspectRatio → 1,
     ImageSize → ImageDimensions[zellealginat]], Show[ImageAdjust[zellealginat, {c, b, γ}], ImageSize → ImageDimensions[zellealginat]]}}],
 {{c, 2., "c"}, 0., 10., .1, Appearance → "Labeled"}, {{b, 0.1, "b"}, 0., 1., .1, Appearance → "Labeled"},
 {{γ, .8, "γ"}, .1, 2, .1, Appearance → "Labeled"}, SaveDefinitions → True]
```



Ergänzungen: Alpha-Kanal

farbfläche

```
Clear[AlphaChannelQ];
AlphaChannelQ[image_Image] := If[RemoveAlphaChannel[image] == image, False, True];
AlphaChannelQ[image_Image3D] := If[RemoveAlphaChannel[image] == image, False, True];
```

```
AlphaChannelQ@farbfläche
```

```
True
```

ColorSeparate@farbfläche



ImageColorSpace@farbfläche

RGB

ImageType@farbfläche

Byte

ImageDimensions@farbfläche

{480, 480}

Vorgriff : Digitale Bildrepräsentation

Diskrete Cosinustransformation (DCT) : reellwertige, diskrete, lineare, orthogonale Transformation

1 D – Hintransformation : $\hat{f}_u = \sum_{m=0}^{M-1} f_m \, \text{Cos}\left[\frac{\pi \left(m + \frac{1}{2}\right) u}{M}\right]$ mit $u = 0 \ldots M - 1$

1 D – Rücktransformation : $f_m = \frac{1}{M}\left(\hat{f}_0 + 2 \sum_{u=1}^{M-1} \hat{f}_u \, \text{Cos}\left[\frac{\pi \left(m + \frac{1}{2}\right) u}{M}\right]\right)$ mit $m = 0 \ldots M - 1$

2 D – Hintransformation : $\hat{f}_{u,v} = \sum_{m=0}^{M-1} \sum_{n=0}^{N-1} f_{m,n} \text{Cos}\left[\frac{\pi \left(m + \frac{1}{2}\right) u}{M}\right] \text{Cos}\left[\frac{\pi \left(n + \frac{1}{2}\right) v}{N}\right]$ mit $u = 0 \ldots M - 1$ und $v = 0 \ldots N - 1$

2 D – Rücktransformation : $f_{m,n} =$

$$\frac{1}{MN}\left(\hat{f}_{0,0} + 2 \sum_{u=1}^{M-1} \sum_{v=1}^{N-1} \hat{f}_{u,v} \text{Cos}\left[\frac{\pi \left(m + \frac{1}{2}\right) u}{M}\right] \text{Cos}\left[\frac{\pi \left(n + \frac{1}{2}\right) v}{N}\right]\right)$$ mit $m = 0 \ldots M - 1$ und $n = 0 \ldots N - 1$

```
mm = 8;
MatrixForm[{Table[DiscretePlot[Cos[π (m + 1 / 2) u / mm], {m, 0, mm - 1}, PlotMarkers → None,
    Ticks → None, PlotRange → {{-1, mm}, {-1, 1}}, ExtentSize → .8, PlotLabel → u], {u, Range[0, mm - 1]}]}]
```

```
mm = 8;
nn = 8;
MatrixForm[Table[DiscretePlot3D[Cos[π (m + 1 / 2) u / mm] * Cos[π (n + 1 / 2) v / nn], {m, 0, mm - 1},
    {n, 0, nn - 1}, PlotMarkers → None, Ticks → None, PlotRange → {{-1, mm}, {-1, nn}, {-1, 1}}, ExtentSize → .8,
    PlotLabel → {u, v}, BoxRatios → {mm + 1, nn + 1, 2}, RotationAction → "Clip"], {u, Range[0, mm - 1]}, {v, Range[0, nn - 1]}]]
```